```
AppendTo[$Path, ToFileName[NotebookDirectory[] <> "\\Hazik"]];

SetDirectory[NotebookDirectory[]];
SetOptions[#, AxesStyle → Arrowheads[Automatic]] & /@
  {Plot, ListPlot, ParametricPlot, ListLinePlot};
LaunchKernels[]

{KernelObject[1, local], KernelObject[2, local]}

Needs["ReactionKinetics`"]

Needs["ReplaceVariables`"]
```

Mathematica

Megoldás, vetület, trajektória

```
log[k_] := Show[Plot[Evaluate[{1, x[t] /.
       NDSolve[{x'[t] == k x[t] (1 - x[t]), x[0] == 0.1}, x[t], {t, 0, 40}][[1]]}],
   {t, 0, 6}, AxesLabel → {t, x[t]}, AxesOrigin → {0, 0}],
  ParametricPlot[{-k x (1 - x), x}, {x, -0.5, 1.5}, AxesOrigin → {0, 0},
   PlotStyle → Directive[Thickness[0.01], Red]],
  PlotRange → {{-1, 6}, {-0.8, 2}}, Prolog →
   {Thick, Brown, Arrow[{{0, -0.1}, {0, -0.5}}], Arrow[{{0, 0.1}, {0, 0.9}}],
    Arrow[{{0, 1.8}, {0, 1.1}}], Green, Arrow[{{2.4, 0.42}, {2.7, 0.485}}]},
  Epilog → {Text[Style["f(x)", Bold, 18], {-0.3, -0.2}, {1, 0}, {0, 1}]}]
log[
 0.8]
```



```
nds = NDSolve[{x'[t] == y[t], y'[t] == -Sin[x[t]], x[0] == 0, y[0] == 1},
    {x, y}, {t, 0, 15}][[1]];
ParametricPlot3D[Evaluate[{{0, x[t], y[t]}, {t, x[t], -1.5},
    {t, 1.5, y[t]}, {t, x[t], y[t]}} /. nds], {t, 0, 10},
 PlotStyle → {Directive[Thick, Brown], Directive[Thick, Green, Dashed],
   Directive[Thick, Blue, Dashed], Directive[Thick, Red]},
 BoxRatios → {4, 1, 1}, AxesLabel → {"t", "x(t)", "y(t)"},
 PlotLabel → "Az inga egyenletének megoldása,\na megoldás
    koordinátafüggvényei  és\n trajektóriája a fázissíkon"]
```



Szimbolikus módszerek

Zárt alakú, szimbolikus megoldás?

```
Together[FunctionExpand[DSolve[{y'[x] == x^2 + y[x]^2}, y[x], x]]]
```

$$\left\{\left\{y[x] \to \frac{-x\,\text{BesselJ}\left[-\frac{3}{4}, \frac{x^2}{2}\right] + x\,\text{BesselJ}\left[\frac{3}{4}, \frac{x^2}{2}\right] C[1]}{\text{BesselJ}\left[\frac{1}{4}, \frac{x^2}{2}\right] + \text{BesselJ}\left[-\frac{1}{4}, \frac{x^2}{2}\right] C[1]}\right\}\right\}$$

Az elemi függvény fogalmának csak történeti jelentősége van. (HF: Program arra, h elemi-e egy fv?)

Közelítő (szimbolikus és numerikus) módszerek

Konkrét eljárások: a bizonyításokból

A fokozatos közelítés módszere

```
F = #1 #2 &;
```

```
A[φ_] := Function[t, -1 + ∫_2^t F[s, φ[s]] ds]
  (*Operátor: függvényhez függvényt rendel*)
```

```
NestList[A, 1 &, 3]
```

```
{1 &, Function[t$, -1 + ∫_2^t$ F[s, (1 &)[s]] ds],
 Function[t$, -1 + ∫_2^t$ F[s, Function[t$, -1 + ∫_2^t$ F[s, (1 &)[s]] ds][s]] ds],
 Function[t$, -1 + ∫_2^t$ F[s, Function[t$,
      -1 + ∫_2^t$ F[s, Function[t$, -1 + ∫_2^t$ F[s, (1 &)[s]] ds][s]] ds][s]] ds]}
```

```
Through[NestList[A, -1 &, 3][z]]
```

$$\left\{-1,\ 1-\frac{z^2}{2},\ -1+\frac{z^2}{2}-\frac{z^4}{8},\ \frac{1}{3}-\frac{z^2}{2}+\frac{z^4}{8}-\frac{z^6}{48}\right\}$$

```
ClearAll[pontos, y, x];
pontos = DSolve[{y'[x] == x y[x], y[2] == -1}, y[x], x][[1]]
```

$$\left\{y[x]\to -e^{-2+\frac{x^2}{2}}\right\}$$

```
Append[Through[NestList[A, -1 &, 3][z]], y[x] /. pontos[[1]]]
```

$$\left\{-1,\ 1-\frac{z^2}{2},\ -1+\frac{z^2}{2}-\frac{z^4}{8},\ \frac{1}{3}-\frac{z^2}{2}+\frac{z^4}{8}-\frac{z^6}{48},\ -e^{-2+\frac{x^2}{2}}\right\}$$

```
Plot[Evaluate[Append[Through[NestList[A, -1 &, 2][x]], pontos]],
  {x, 2, 4}, PlotLegends → "Expressions"]
```



Jó esetben megsejtjük az általános alakot, majd bebizonyítjuk, megvizsgálva a konvergenciatartományát.

Megoldás hatványsor alakjában

```
y[x_] = 1 + Sum[a[i] x^i, {i, 3}] + O[x]^4
```

$$1 + a[1]\,x + a[2]\,x^2 + a[3]\,x^3 + O[x]^4$$

```
D[y[x], x]^2 - y[x] == x
```

$\left(-1 + a[1]^2\right) + (-a[1] + 4\, a[1]\, a[2])\, x + \left(-a[2] + 4\, a[2]^2 + 6\, a[1]\, a[3]\right) x^2 + O[x]^3 == x$

Ez egy nagyon ügyes függvény!

```
LogicalExpand[%]
```

$-1 + a[1]^2 == 0 \,\&\&\, -1 - a[1] + 4\, a[1]\, a[2] == 0 \,\&\&\, -a[2] + 4\, a[2]^2 + 6\, a[1]\, a[3] == 0$

```
Solve[%]
```

$\left\{\{a[1] \to -1,\ a[2] \to 0,\ a[3] \to 0\},\ \left\{a[1] \to 1,\ a[2] \to \frac{1}{2},\ a[3] \to -\frac{1}{12}\right\}\right\}$

```
y[x] /. %
```

$\left\{1 - x + O[x]^4,\ 1 + x + \frac{x^2}{2} - \frac{x^3}{12} + O[x]^4\right\}$

```
Normal[%]
```

$\left\{1 - x,\ 1 + x + \frac{x^2}{2} - \frac{x^3}{12}\right\}$

Miért kaptunk két megoldást?

Jó esetben megsejtjük az általános alakot (a program is segíthet), majd bebizonyítjuk, megvizsgálva a konvergenciatartományát.

Egy hasznos függvény: ParametricNDSolve

Úgy bánik a numerikus megoldással, mintha szimbolikus lenne!

```
sol = ParametricNDSolve[
  {x''[t] + a x[t] == 0, x[0] == 1, x'[0] == 0}, {x}, {t, 0, 10}, {a}]
```

{x → ParametricFunction[Expression: x Parameters: {a}]}

```
Plot[Evaluate[x[a][10] /. sol], {a, 0, 2}, AxesLabel → {"a", "x[a][10]"},
 PlotLabel → "Nem az idő függvénye,\n a paraméteré!"]
```



Az ábra alapján tudunk jó kezdeti becsléseket kapni, ez kell a FindRootnak.

```
FindRoot[x[a][10] /. sol, {a, #}] & /@ {0, 0.2, 0.5, 1, 2}
```

```
{{a → 0.024674}, {a → 0.222066}, {a → 0.61685}, {a → 1.20903}, {a → 1.99859}}
```

Első házi feladatsor

3. feladat (Az Euler-módszer)

A pontos megoldás

```
ClearAll[pontos, y, x];
pontos = DSolve[{y'[x] == 1 - y[x]/x, y[2] == -1}, y[x], x][[1]]
```

$$\left\{y[x] \to \frac{-8 + x^2}{2\,x}\right\}$$

```
NestList[ff, x0, 2]
```

```
{x0, ff[x0], ff[ff[x0]]}
```

Ez jó lesz nekünk!

```
euler[f_, h_, x0_, y0_, n_: 3] :=
 Module[{lep, x, y}, lep[{x_, y_}] := {x + h, y + h f[x, y]};
  NestList[lep, {x0, y0}, n]]
```

```
ClearAll[F];
F[x_, y_] := 1 - y/x;
```

```
lp = ListLinePlot[euler[F, 0.5, 2, -1, 3],
  PlotStyle → {RGBColor[1, 0, 0]}, PlotLabel → "közelítő"]
```



```
pont = Plot[y[x] /. pontos, {x, 2, 3.5}, PlotLabel → "pontos"]
```




```
Show[lp, pont, PlotLabel → "Összevetés"]
```




```
TableForm[euler[F, 0.5, 2, -1, 3], TableHeadings → {None, {"xᵢ", "yᵢ"}}]
```

x_i	y_i
2	-1
2.5	-0.25
3.	0.3
3.5	0.75

4. feladat

```
ds = DSolveValue[{y'[x] == x y[x], y[1] == -1}, y[x], x];
```

```
StreamPlot[{1, x y}, {x, -2, 3}, {y, -5, 4}, PlotLabel → "Zöld: közelítő\n",
 Epilog → First@Plot[ds, {x, -2, 2}, PlotStyle → {Thickness[0.01]}], Prolog →
  {Red, PointSize[0.05], Point[{1, -1}], Blue, Point[{2, Evaluate[ds /. x → 2]}],
   Green, Point[Last[euler[#1 #2 &, 0.2, 1, -1, 5]]]}]
```




```
Last[euler[#1 #2 &, 0.2, 1, -1, 5]]
```

```
{2., -3.41921}
```

5. feladat (fázisegyenes)

```
Show[Plot[Evaluate[{-1, 1,
    y[x] /. NDSolve[{y'[x] == y[x]^2 - 1, y[0] == -0.1}, y[x], {x, 0, 40}][[1]]}],
  {x, 0, 3}, AxesLabel → {x, y[x]}, AxesOrigin → {0, 0}],
 ParametricPlot[{-y^2 + 1, y}, {y, -1.5, 1.5}, AxesOrigin → {0, 0},
  PlotStyle → Directive[Thickness[0.01], Red]],
 PlotRange → {{-1, 3}, {-1.6, 2}}, Prolog →
  {Thick, Brown, Arrow[{{0, -1.7}, {0, -1.1}}], Arrow[{{0, 0.9}, {0, -0.9}}],
   Arrow[{{0, 1.1}, {0, 1.8}}]},
 Epilog -> Text[Style["f(x)", Bold, 16], {-0.7, -0.8}, {1, 0}, {0, 1}]]
```



6. feladat (egzakttá tehető, de homogén is)

$$eq6 = y'[x] == \frac{x^2 - 2 x y[x] - 3 y[x]^2}{-3 x^2 - 2 x y[x] + y[x]^2};$$

```
DSolve[eq6, y[x], x]
```

$$\left\{\left\{y[x] \to \frac{1}{2}\left(-e^{C[1]} - 2 x - e^{\frac{C[1]}{2}} \sqrt{e^{C[1]} + 8 x}\right)\right\}, \left\{y[x] \to \frac{1}{2}\left(-e^{C[1]} - 2 x + e^{\frac{C[1]}{2}} \sqrt{e^{C[1]} + 8 x}\right)\right\}\right\}$$

```
rv6 = ReplaceVariables[eq6, ξ → x, y[x] → ξ η[ξ], η[ξ]]
```

$$-1 + \eta[\xi]^2 - 3 \xi \eta'[\xi] + \xi \eta[\xi] \eta'[\xi] == 0$$

```
eq66 = Equal @@@ (Solve[rv6, η'[ξ]][[1]])
```

$$\left\{\eta'[\xi] == \frac{1 - \eta[\xi]^2}{\xi (-3 + \eta[\xi])}\right\}$$

```
DSolve[eq66, η[ξ], ξ]
```

$$\left\{\left\{\eta[\xi] \to \frac{-e^{C[1]} - 2 \xi - e^{\frac{C[1]}{2}} \sqrt{e^{C[1]} + 8 \xi}}{2 \xi}\right\}, \left\{\eta[\xi] \to \frac{-e^{C[1]} - 2 \xi + e^{\frac{C[1]}{2}} \sqrt{e^{C[1]} + 8 \xi}}{2 \xi}\right\}\right\}$$

Szétválasztható változójú lett.

7. feladat (Euler)

```
ReplaceVariables[x^2 y''[x] - 4 x y'[x] + 6 y[x] == 0, ξ → Log[x], y[x] → η[ξ], η[ξ]]
```

```
6 η[ξ] - 5 η′[ξ] + η″[ξ] == 0
```

Az együtthatók lehetnek *x* lineáris függvényei is lehetnek, az egyenlet magasabbrendű is lehet.

10. feladat

Először fölírjuk a görbesereg differenciálegyenletét, utána pedig megrajzoljuk az ortogonális trajektóriákat.

```
Eliminate[{y[x] == c x^2, y'[x] == D[c x^2, x]}, c]
```

```
x y′[x] == 2 y[x]
```

```
Solve[%, y'[x]]
```

$$\left\{\left\{y'[x] \to \frac{2\,y[x]}{x}\right\}\right\}$$

```
StreamPlot[{{1, 2 y/x}, {1, -x/(2 y)}}, {x, -2, 2}, {y, -2, 2}]
```



Második házi feladatsor

Ezt többször fogjuk használni, de vigyázat, nem feltételek nélkül működik.

```
torhs[eq_] := Last /@ eq /. u_[t] → u
```

2. feladat

```
DSolveValue[{x'[t] == y[t] + ℯ^(2 t)/t, y'[t] == -x[t] + Tan[t]}, {x[t], y[t]}, t]
```

$$\{C[1]\,\mathrm{Cos}[t] + C[2]\,\mathrm{Sin}[t] + \left(-\mathrm{Cos}[t] - \frac{1}{2}\,i\,(-\mathrm{ExpIntegralEi}[(2-i)\,t] + \mathrm{ExpIntegralEi}[(2+i)\,t])\right)\mathrm{Sin}[t] + \mathrm{Cos}[t]\left(\frac{1}{2}\,(\mathrm{ExpIntegralEi}[(2-i)\,t] + \mathrm{ExpIntegralEi}[(2+i)\,t]) + \mathrm{Log}\left[\mathrm{Cos}\left[\frac{t}{2}\right] - \mathrm{Sin}\left[\frac{t}{2}\right]\right] - \mathrm{Log}\left[\mathrm{Cos}\left[\frac{t}{2}\right] + \mathrm{Sin}\left[\frac{t}{2}\right]\right] + \mathrm{Sin}[t]\right),\ C[2]\,\mathrm{Cos}[t] + \mathrm{Cos}[t]\left(-\mathrm{Cos}[t] - \frac{1}{2}\,i\,(-\mathrm{ExpIntegralEi}[(2-i)\,t] + \mathrm{ExpIntegralEi}[(2+i)\,t])\right) - C[1]\,\mathrm{Sin}[t] - \mathrm{Sin}[t]\left(\frac{1}{2}\,(\mathrm{ExpIntegralEi}[(2-i)\,t] + \mathrm{ExpIntegralEi}[(2+i)\,t]) + \mathrm{Log}\left[\mathrm{Cos}\left[\frac{t}{2}\right] - \mathrm{Sin}\left[\frac{t}{2}\right]\right] - \mathrm{Log}\left[\mathrm{Cos}\left[\frac{t}{2}\right] + \mathrm{Sin}\left[\frac{t}{2}\right]\right] + \mathrm{Sin}[t]\right)\}$$

Kézzel is ilyen csúnya lett?

3. feladat

```
eq3 = {x'[t] == 3 y[t] - 3 x[t],
   y'[t] == (1 + a^2) x[t] - y[t] - x[t] z[t], z'[t] == x[t] y[t] - z[t]};
```

```
rhs3 = torhs[eq3]
```

$$\{-3\,x + 3\,y,\ (1+a^2)\,x - y - x\,z,\ x\,y - z\}$$

Egyensúlyi helyzetek:

```
sol3 = Solve[rhs3 == 0, {x, y, z}]
```

$$\{\{x \to 0,\ y \to 0,\ z \to 0\},\ \{x \to -a,\ y \to -a,\ z \to a^2\},\ \{x \to a,\ y \to a,\ z \to a^2\}\}$$

```
MatrixForm[J = D[rhs3, {{x, y, z}, 1}]]
```

$$\begin{pmatrix} -3 & 3 & 0 \\ 1 + a^2 - z & -1 & -x \\ y & x & -1 \end{pmatrix}$$

```
Eigenvalues /@ (J /. sol3)
```

$$\{\{-1,\ -2-\sqrt{4+3\,a^2},\ -2+\sqrt{4+3\,a^2}\},\ \{\mathrm{Root}[6\,a^2 + (4+a^2)\,\#1 + 5\,\#1^2 + \#1^3\,\&,\,1],\ \mathrm{Root}[6\,a^2 + (4+a^2)\,\#1 + 5\,\#1^2 + \#1^3\,\&,\,2],\ \mathrm{Root}[6\,a^2 + (4+a^2)\,\#1 + 5\,\#1^2 + \#1^3\,\&,\,3]\},\ \{\mathrm{Root}[6\,a^2 + (4+a^2)\,\#1 + 5\,\#1^2 + \#1^3\,\&,\,1],\ \mathrm{Root}[6\,a^2 + (4+a^2)\,\#1 + 5\,\#1^2 + \#1^3\,\&,\,2],\ \mathrm{Root}[6\,a^2 + (4+a^2)\,\#1 + 5\,\#1^2 + \#1^3\,\&,\,3]\}\}$$

```
Simplify[-CharacteristicPolynomial[#, λ] & /@ (J /. sol3)]
```

$$\{(1+\lambda)\,(-3\,a^2 + \lambda\,(4+\lambda)),\ a^2\,(6+\lambda) + \lambda\,(4 + 5\,\lambda + \lambda^2),\ a^2\,(6+\lambda) + \lambda\,(4 + 5\,\lambda + \lambda^2)\}$$

Az origó mindig instabilis, a többi aszimptotikus stabilitásához elegendő, ha a megfelelő

```
Expand[a^2 (6 + λ) + λ (4 + 5 λ + λ^2)]
```

$6 a^2 + 4 \lambda + a^2 \lambda + 5 \lambda^2 + \lambda^3$

polinom nemcsak a Stodola-kritériumot teljesíti (az együtthatók pozitívak), hanem a Routh-Hurwitz-kritériumot is, ami itt egyszerűen az alábbi:

```
Reduce[5 (4 + a^2) > 6 a^2, a]
```

$-2\sqrt{5} < a < 2\sqrt{5}$

```
a = 2 √5 ; Solve[6 a^2 + (4 + a^2) #1 + 5 #1^2 + #1^3 &[x] == 0, x]
```

$\{\{x \to -5\}, \{x \to -2 i \sqrt{6}\}, \{x \to 2 i \sqrt{6}\}\}$

```
Manipulate[Plot[6 a^2 + (4 + a^2) #1 + 5 #1^2 + #1^3 &[x], {x, -6, 4}],
 {{a, -2 √5 }, -5, 5, 0.5}]
```



Mi a helyzet a határon, ahol a linearizálás nem segít?

4. feladat

```
eq4 = {x'[t] == 3 x[t] - y[t] - 2 z[t],
   y'[t] == -8 x[t] + 6 y[t] + 10 z[t], z'[t] == 5 x[t] - 3 y[t] - 5 z[t]};
```

```
rhs4 = torhs[eq4]
```

{3 x − y − 2 z, −8 x + 6 y + 10 z, 5 x − 3 y − 5 z}

Az együtthatóátrixot így nyerhetjük ki:

```
cr = CoefficientRules[rhs4]
```

```
{{{1, 0, 0} → 3, {0, 1, 0} → -1, {0, 0, 1} → -2},
 {{1, 0, 0} → -8, {0, 1, 0} → 6, {0, 0, 1} → 10},
 {{1, 0, 0} → 5, {0, 1, 0} → -3, {0, 0, 1} → -5}}
```

```
coeffmat = Map[Last, cr, {2}]
```

```
{{3, -1, -2}, {-8, 6, 10}, {5, -3, -5}}
```

```
Eigenvalues[coeffmat]
```

```
{2, 1, 1}
```

5. feladat

Egy sorozat ábra jön

```
StreamPlot[{x - y, y - 4 x}, {x, -3, 3}, {y, -3, 3},
 Epilog → {Red, PointSize -> Large, Point[{0, 0}]}, PlotTheme → "Scientific"]
```



```
Table[StreamPlot[{x - y, y - 4 x}, {x, -3, 3}, {y, -3, 3}, StreamScale → None,
  PlotLabel → p, StreamPoints → p], {p, {Coarse, Medium, Automatic, Fine}}]
```




```
StreamPlot[{x - y, y - 4 x}, {x, -3, 3}, {y, -3, 3},
 StreamPoints → {{{{1, 0}, Red}, {{-1, -1}, Green}, Automatic}}]
```



```
StreamPlot[{x - y, y - 4 x}, {x, -3, 3},
 {y, -3, 3}, VectorPoints → 10, VectorStyle → Red]
```



6. feladat

```
eq6 = {x'[t] == -7 Sinh[x[t]] + 2 y[t],
   y'[t] == 2 Sin[x[t]] - 6 y[t] + 2 z[t], z'[t] == 2 (x[t] + 1) y[t] - 5 z[t]};
```

```
rhs6 = torhs[eq6];
```

```
rhs6 /. {x → 0, y → 0, z → 0}
```

```
{0, 0, 0}
```

Van-e másik egyensúlyi helyzet?

```
eli = Eliminate[Thread[rhs6 == 0], {y, z}]
```

```
(91 - 14 x) Sinh[x] == 10 Sin[x]
```

```
List @@ eli
```

```
{(91 - 14 x) Sinh[x], 10 Sin[x]}
```

```
Plot[Evaluate[List @@ eli], {x, -3, 3},
 Epilog → Rectangle[{-2.6, 20}, {-0.5, 450}, Plot[Evaluate[List @@ eli],
    {x, -0.1, 0.1}, Frame → True, PlotLabel → "Közelebbről:"]]]
```



Belátható, hogy csak egy metszéspontjuk van.

```
D[rhs6, {{x, y, z}, 1}] /. Thread[{x, y, z} → 0]
```

```
{{-7, 2, 0}, {2, -6, 2}, {0, 2, -5}}
```

```
Eigenvalues[%]
```

```
{-9, -6, -3}
```

```
VectorPlot3D[rhs6, {x, -2, 2}, {y, -3, 3}, {z, -4, 4}, PlotTheme → "Detailed"]
```



7. feladat

```
eq7 = {-2 x - y, (3 + b) x - 5 y};
```

```
Plot[τ^2 / 4, {τ, -3, 3}, PlotRange → {-3, 3}, AxesLabel → {"nyom: τ", "determ: δ"}]
```




```
cr = CoefficientRules[eq7, {x, y}]
```

```
{{{1, 0} → -2, {0, 1} → -1}, {{1, 0} → 3 + b, {0, 1} → -5}}
```

```
A = Map[Last, #] & /@ cr
```

```
{{-2, -1}, {3 + b, -5}}
```

```
{Tr[A], Det[A]}
```

```
{-7, 13 + b}
```

```
Plot[τ^2 / 4, {τ, -10, 10}, PlotRange → {-20, 50},
 AxesLabel → {"nyom: τ", "determ: δ"},
 Epilog → {Thick, Red, Arrow[{{-7, -20}, {-7, 40}}], Black,
   Text["A b paraméter hatása", {-7, 45}], Text[Style["Nyereg, IS", 14, Blue],
    {0, -15}], Text[Style["S csomó", 14, Blue], {-8, 8}],
   Text[Style["S centrum", 14, Black], {-4, 12}],
   Text[Style["S fókusz", 14, Blue], {-3, 30}],
   Text[Style["τ^2=4δ", 14, Brown], {8, 20}], Circle[{-7, 3 + 37/4}, 0.5]}]
```




```
Solve[(-7)^2 == 4 (3 + b), b]
```

8. feladat

```
eq8 = {x'[t] == x[t]^3 + x[t] y[t]^2 - x[t] - y[t], y'[t] == y[t]^3 + y[t] x[t]^2 - y[t] + x[t]};
```

```
rhs8 = torhs[eq8];
```

```
StreamPlot[rhs8, {x, -2, 2}, {y, -2, 2}, Epilog → {Red, Thick, Circle[]}]
```




```
Eigenvalues[D[rhs8, {{x, y}, 1}] /. {x → 0, y → 0}]
```

{-1 + ⅈ, -1 - ⅈ}

```
Solve[rhs8 == 0, {x, y}]
```

{{x → 0, y → 0}}

Áttérünk polárkoordinátákra

```
ClearAll[x, y, φ, t, ρ];
```

```
helyett = {x[t] → ρ[t] Cos[φ[t]], y[t] → ρ[t] Sin[φ[t]]}
```

{x[t] → Cos[φ[t]] ρ[t], y[t] → Sin[φ[t]] ρ[t]}

```
dhelyett = D[helyett, t]
```

{x′[t] → Cos[φ[t]] ρ′[t] - Sin[φ[t]] ρ[t] φ′[t],
 y′[t] → Sin[φ[t]] ρ′[t] + Cos[φ[t]] ρ[t] φ′[t]}

```
Equal @@@ dhelyett
```

{x′[t] == Cos[φ[t]] ρ′[t] - Sin[φ[t]] ρ[t] φ′[t],
 y′[t] == Sin[φ[t]] ρ′[t] + Cos[φ[t]] ρ[t] φ′[t]}

```
soleqd = Solve[Equal @@@ dhelyett, {ρ'[t], φ'[t]}][[1]] // Simplify
```

$$\left\{\rho'[t] \to \text{Cos}[\varphi[t]]\, x'[t] + \text{Sin}[\varphi[t]]\, y'[t],\ \varphi'[t] \to \frac{\text{Sin}[\varphi[t]]\,(-x'[t] + \text{Cot}[\varphi[t]]\, y'[t])}{\rho[t]}\right\}$$

```
pill = soleqd /. Rule @@@ eq8 /. helyett // Simplify
```

$$\left\{\rho'[t] \to \rho[t]\left(-1 + \rho[t]^2\right),\ \varphi'[t] \to 1\right\}$$

```
pill /. Rule → Equal
```

$$\left\{\rho'[t] == \rho[t]\left(-1 + \rho[t]^2\right),\ \varphi'[t] == 1\right\}$$

```
Equal @@@ pill
```

$$\left\{\rho'[t] == \rho[t]\left(-1 + \rho[t]^2\right),\ \varphi'[t] == 1\right\}$$

Innen látszik, hogy az egységkörvonal instabilis mindkét oldalról.

Szerkesszünk Bendixson-zsákot.

9. Feladat

```
eq9[μ_: 0] := {x'[t] == y[t], y'[t] == x[t] - x[t]^3 - μ (y[t]^2 - 2 x[t]^2 + x[t]^4)};
```

```
eq9[]
```

$$\left\{x'[t] == y[t],\ y'[t] == x[t] - x[t]^3\right\}$$

```
rhs9[μ_: 0] := torhs[eq9[μ]]
```

```
rhs9[]
```

$$\left\{y,\ x - x^3\right\}$$

Vegyük észre :)

```
Map[Integrate[#, t] &, y[t] y'[t] == (x[t] - x[t]^3) x'[t]]
```

$$\frac{y[t]^2}{2} == \frac{x[t]^2}{2} - \frac{x[t]^4}{4}$$

Tehát egy első integrál

```
φ[x_, y_] := 2 y^2 + x^4 - 2 x^2
```

És valóban

```
D[φ[x, y], {{x, y}, 1}].rhs9[] // Expand
```

0

```
sp9 = StreamPlot[rhs9[], {x, -3, 3}, {y, -3, 3}]
```




```
cp9 = ContourPlot[φ[x, y], {x, -3, 3}, {y, -3, 3}]
```



```
Show[cp9, sp9, PlotLabel → "Fordítva nem látszik!"]
```




```
D[φ[x, y], {{x, y}, 1}].rhs9[μ] // Factor
```

$-4\,y\,(-2\,x^2 + x^4 + y^2)\,\mu$

```
φ[x, y]
```

$-2\,x^2 + x^4 + 2\,y^2$

```
NSolve[rhs9[10] == 0, {x, y}, Reals]
```

```
{{y → 0, x → -1.44035}, {y → 0, x → -0.0499376}, {y → 0, x → 0}, {y → 0, x → 1.39029}}
```

```
NSolve[rhs9[0] == 0, {x, y}, Reals]
```

```
{{x → -1., y → 0}, {x → 0, y → 0}, {x → 1., y → 0}}
```

```
NSolve[rhs9[-2] == 0, {x, y}, Reals]
```

```
{{y → 0, x → -1.3131}, {y → 0, x → 0}, {y → 0, x → 0.242431}, {y → 0, x → 1.57067}}
```

```
Manipulate[Plot[x - x^3 + μ x^2 (2 - x^2), {x, -2, 2},
  PlotLabel → "Változik a gyökök száma"], {μ, -2, 6, 0.1}]
```




```
Manipulate[StreamPlot[rhs9[μ], {x, -3, 3}, {y, -3, 3}, PerformanceGoal → "Speed"],
 {{μ, 1}, -1, 14, 0.2}]
```



```
StreamPlot[rhs9[2], {x, -3, 3}, {y, -3, 3}, Prolog → First@
   ContourPlot[φ[x, y], {x, -3, 3}, {y, -3, 3}, Contours → {0.01, 0.1, 1, 5, 10}]]
```



10. feladat

```
eq10 = {x'[t] == x[t] - y[t] - x[t]^3, y'[t] == x[t] + y[t] - y[t]^3};
```

```
rhs10 = torhs[eq10]
```

$\{x - x^3 - y, x + y - y^3\}$

```
Eigenvalues[D[rhs10, {{x, y}, 1}] /. {x → 0, y → 0}]
```

$\{1 + i, 1 - i\}$

Instabil fókusz.

```
sp = StreamPlot[rhs10, {x, -3, 3}, {y, -3, 3}]
```




```
V[x_, y_] := x^2 + y^2
```

```
D[V[x, y], {{x, y}, 1}].rhs10 // Simplify
```

$-2\left(-x^2+x^4-y^2+y^4\right)$

```
RegionPlot[-2 (-x^2 + x^4 - y^2 + y^4) < 0,
 {x, -3, 3}, {y, -3, 3}, PlotLabel → "Hol negatív?"]
```



```
cp = ContourPlot[V[x, y], {x, -3, 3}, {y, -3, 3}, ContourShading → None,
    Contours → Range[0, 10., 1.0], ContourStyle → Directive[Red, Thick]];
```

```
Show[sp, cp]
```



Megint segíthet a Bendixson-zsák.

További érdekes példák, alkalmazások

Hővezetés

```
ho = NDSolve[{D[u[t, x], t] == D[u[t, x], {x, 2}], u[t, 0] == 0,
    u[0, x] == Sin[x], u[t, 2 π] == 0}, u, {t, 0, 10}, {x, 0, 2 π}]
```

```
{{u → InterpolatingFunction[ Domain: {{0., 10.}, {0., 6.28}}
                             Output: scalar ]}}
```

```
Plot3D[Evaluate[u[t, x] /. ho], {t, 0, 10}, {x, 0, 2 π}, PlotRange → All,
 ColorFunction → "TemperatureMap", AxesLabel → {t, x, "u(t,x)"}]
```



Kémiai reakciókinetika

Robertson-reakció

```
ClearAll[a, b, c, r1, r2, r3, t];
{r1, r2, r3} = {k1 a[t], k2 b[t]^2, k3 b[t] c[t]};
eqns = {a'[t] == -r1 + r3, b'[t] == r1 - r2 - r3};

eqEqn = {a[t] + b[t] + c[t] == 1};
icEqn = {a[0] == 1, b[0] == 0, c[0] == 0};

params = {k1 → 0.04, k2 → 3 × 10^7, k3 → 10^4};
sol = NDSolve[{eqns, eqEqn, icEqn} /. params, {a, b, c}, {t, 0, 10^8}];

LogLinearPlot[Evaluate[#[t] /. sol],
   {t, 10^-6, 10^8}, PlotRange → All, PlotLabel → #[t]] & /@ {a, b, c}
```



A három összege? Mutassuk meg, hogy tényleg állandó.

Elektromos hálózat




```
components = {r iR[t] == vR[t], l iL'[t] == vL[t], iC[t] == c vC'[t]};
connections =
  {iC[t] + iR[t] + iL[t] == i1[t], v[t] == vR[t], vR[t] == vL[t], vL[t] == vC[t]};

i1[t_] := 1;

ic = {v[0] == vR[0] == vR[0] == vC[0] == iR[0] == iL[0] == iC[0] == 0};
params = {r → 10, c → 10^-3, l → 0.4};
sol = NDSolve[{components, connections, iL[0] == 0, v[0] == 0} /. params,
   {iR, iL, iC, v}, {t, 0, 0.2}, AccuracyGoal → 7];

{Plot[Evaluate[{iR[t], iL[t], iC[t]} /. sol], {t, 0, 0.2}],
 Plot[v[t] /. sol, {t, 0, 0.2}]}
```



Értelmezzük az eredményt.

Rezgőmozgás

```
ClearAll[springGraphics]
springGraphics[r_: 0, OptionsPattern[]] :=
 Module[{w = N[OptionValue[width]],
   l = N[OptionValue[length]], n = OptionValue[segments]},
  Graphics[{White, Thickness[Large],
    Line[-Flatten[{
        {{{0, 0}}},
        Table[{{0, y}, {w, y + (0.8 l + r) / (4 n)}, {-w, y + 3 (0.8 l + r) / (4 n)}},
         {y, 0.05 l, (0.85 l + r) - (0.8 l + r) / n, (0.8 l + r) / n}],
        {{{0, 0.85 l + r}, {0, l + r}}}
       }, 2]],
    Black, EdgeForm[{White, Thick}], Disk[{0, -l - r}, 0.2]
   }]
 ]
Options[springGraphics] = {length → 1, segments → 4, width → 0.1};
solSpring = ParametricNDSolveValue[
   {mass x''[t] + damping x'[t] + mass 4 π^2 freq^2 x[t] == 0, x[0] == x0, x'[0] == v0},
   x, {t, -2, 60}, {x0, v0, mass, damping, freq}];
```

```
solSpr[t_] = solSpring[1, 0, 1, 0.5, 1][t];
GraphicsRow[{
  Plot[{solSpr[t], solSpr'[t]}, {t, 0, 2}, FrameLabel → {t, None},
   PlotLegends → Placed[{"Pozíció", "Sebesség"}, ImageSize → 400]],
  ParametricPlot[{solSpr[t], solSpr'[t]}, {t, 0, 2},
   AspectRatio → 1, FrameLabel → {"Pozíció", "Sebesség"}]
 }, ImageSize → 900, Spacings → 1]
```

Placed::labpos : ImageSize → 400 is not a valid position for the placement of labels. >>

Placed::labpos : ImageSize → 400 is not a valid position for the placement of labels. >>



```
Manipulate[
 Animate[
  Show[
   Graphics[{White, Disk[{0, 0}, 0.05], Line[{{-2, 0}, {2, 0}}],
     Line /@ Table[{{x, 0}, {x - 0.1, -0.1}}, {x, -2, 2, 0.2}]}],
   If[showMeasure, Graphics[{White, Line[{{-2, -2}, {2, -2}}],
      Arrowheads[{-0.03, 0.03}], Arrow[{{-1, -2}, {-1, -2 - solSpr[t]}}],
      Line[{{-1.1, -2 - solSpr[t]}, {0, -2 - solSpr[t]}}],
      Text[Style[x, 18], {-1.1, -2 - solSpr[t] / 2}]}], Graphics[]],
   springGraphics[solSpr[t], length → 2, segments → 16],
   PlotRange → {{-2, 2}, {-3.5, 1}}, Background → Black],
  {t, 0, 60, 0.01}, AnimationRate → 1, AnimationRunning → False]
 , {showMeasure, {False, True}}]
```



Hőmérsélet szabályozás kapcsolgatással: diszkrét

bevatkozás, hibrid rendszer

```
solTemp = NDSolveValue[{x'[t] == -0.5 (x[t] - 30 - 2 s[t]),
   x[0] == 0, s[0] == 1, WhenEvent[x[t] == 30 + s[t], s[t] → -s[t]]},
  x[t], {t, 0, 50}, DiscreteVariables → {s}]
```

```
InterpolatingFunction[ Domain: {{0., 50.}}
                       Output: scalar ][t]
```

```
Plot[{30, solTemp}, {t, 0, 50}, PlotRange → All,
 PlotStyle → {{Black, Dashed}, {Red, Thick}}]
```



Késleltetett egyenlet légzésre

```
Manipulate[
 Module[{sol, x, t},
  sol = First[NDSolve[{x'[t] == a x[t - τ] / (1 + x[t - τ]^10) - b x[t],
      x[t /; t ≤ 0] == 1 / 2}, x, {t, 0, 500}]];
  ParametricPlot[Evaluate[{x[t], x[t - τ]} /. sol], {t, 300, 500},
   PlotRange → {{0, 2}, {0, 2}}]], {{a, .2}, 0, 1}, {{b, .1}, 0, 1},
 {{τ, 17}, 1,
  20}]
```



További apróságok

Definíciók, megoldások:

```
Hyperlink["EqWorld", "http://eqworld.ipmnet.ru/en/solutions/fpde/fpdetoc3.htm"]
```

EqWorld

```
sol = DSolve[
  {x'[t] == 1, x[0] == 0, WhenEvent[{t == 1, t == 2}, x[t] → x[t] + 1]}, x, {t, 0, 3}]
```

$$\left\{\left\{x \to \text{Function}\left[\{t\}, \begin{cases} t & 0 \le t \le 1 \\ 1 + t & 1 < t \le 2 \\ 2 + t & 2 < t \le 3 \\ \text{Indeterminate} & \text{True} \end{cases}\right]\right\}\right\}$$

```
Plot[x[t] /. %, {t, 0, 3}]
```



PDE-rendszer ☺; gradiensből függvény (vö. egzakt egyenletek és Cauchy-Riemann-egyenletek)

```
DSolve[{D[f[x, y], x] == 2 x y^3 + y Cos[x y],
    D[f[x, y], y] == 3 x^2 y^2 + x Cos[x y]}, f[x, y], {x, y}]
```

$$\{\{f[x, y] \to x^2 y^3 + C[1] + \text{Sin}[x\, y]\}\}$$

Hivatkozások

Hirsch, M. W.; Smale, S.; Devaney, R. L. Differential equations, dynamical systems and an introduction to chaos, Elsevier, Academic Press.

```
Hyperlink[hds, "http://www.math.upatras.gr/~bountis/files/def-eq.pdf"]
```

Tóth, J.; Simon, L. P.: Differenciálegyenletek. Bevezetés az elméletbe és az alkalmazásokba, TYPOTEX, Budapest,

```
Hyperlink[ts,
 "http://www.typotex.hu/book/250/toth_janos_simon_peter_differencialegyenletek"
 ]
```

Tóth, J., Csikja, R.; Simon, L. P. : Differenciálegyenletek feladatgyűjtemény.

```
Hyperlink[tcss, "http://tankonyvtar.ttk.bme.hu/pdf/166.pdf"]
```

```
Hyperlink["Stabilis poliomok",
 "https://www.researchgate.net/profile/Janos_Toth4/publication/260458110
   _Stability_of_polynomials/links/0c9605315cb1fefeb5000000.pdf"]
```